SRF005 Ultrasonic Range Sensor

Specification:

The SRF005 ultrasonic range sensor detects objects in it's path and can be used to calculate the range to the object. It is sensitive enough to detect a 3cm diameter broom handle at a distance of over 3m.

Voltage	- 5V
Current	- 30mA Typ. 50mA Max.
Frequency	- 40KHz
Max Range	- 3 m
Min Range	- 3 cm
Sensitivity	- Detect 3cm diameter broom handle at > 3 m
Input Trigger	- 10uS Min. TTL level pulse
Echo Pulse	- Positive TTL level signal, width proportional to range.
Small Size	- 43mm x 20mm x 17mm height

The module can be used in two different modes:

Single Pin	- Single microcontroller pin (08M, and all M2 and X2 parts)
Dual Pin	- Separate PICAXE microcontroller trigger and echo pins

Most users using the latest generation (M2 and X2) PICAXE parts should select 'single pin' connection mode.

Single Pin Connection Mode:

The PICAXE-08M and all M2/X2 parts have bi-directional pins, so the SRF005 can connect to a single i/o pin.

There are two way to achieve this connection on the SRF005, via the 5 way header or via the 3 way header. The 3 way header is designed to be compatible with 'servo extension leads' (e.g. part DAG001) so is often the preferred method on new designs. The 5 way header is compatible with older SRF005 modules/PCBs.

Using the 5 way header (note +5V and 0V are marked on the SRF005):

+5V	Connect to 5V
Not used	Do not connect
Signal	Connect directly to the PICAXE pin
Mode	Connect to 0V
0V	Connect to 0V

Using the 3 way header (note SIG and 0V are marked on the SRF005):

Signal (SIG)	Connect directly to the PICAXE pin
+5V	Connect to 5V
0V	Connect to 0V

When using the 3 pin header you MUST also solder a wire link between the mode and 0V on the 5 way header (ie a wire link between pads 4 and 5 on the 5 way header).

Take care not to overheat, and therefore damage, the solder connection pads whilst making connections.

Example PICAXE Program 1:

The following program give an example of how to use the SRF005 module with a PICAXE microcontroller in single pin mode. The special 'ultra' command is designed for use with the SRF005 in single pin mode.

```
symbol SIG = C.1              ; Define pin for Trigger & Echo (All M2, X2 parts)
symbol range = w1             ; 16 bit word variable for range

main:
    ultra SIG,range           ; use dedicated ultra command
    debug range               ; display range via debug command
    pause 50                  ; short delay
    goto main                 ; loop around forever
```

Example Logicator Flowsheet:

The following flowchart give an example of how to use the SRF005 module with a PICAXE microcontroller in single pin mode. The special 'ultra' cell is designed for use with the SRF005.



Technical Details (Single Pin Mode):

The input/Output pin is used to trigger the SRF005 module via a 'pulsout' command and then the pin is converted to an input. The SRF005 module then sends out the sonic burst, and sets the pin high for the time it takes the sonic burst to be returned. Therefore the same PICAXE pin is then used to receive and time this echo pulse via a 'pulsin' command.



The length of the echo pulse is then divided by 5.8 to give a value in cm, and displayed on the computer screen via the 'debug' command. Note that a word variable, w1, is used for the echo timing, as the echo pulse may be a value greater than 255 (maximum value of a byte variable). Word variables are made up of two byte variables and so have a maximum value of 65535 (in this case w1 is made up of b2 and b3, so these two byte variables must not be used anywhere else in the program).

Example Single Pin PICAXE Program 2:

```
symbol SIG = C.1              ; Define pin for Trigger & Echo (All M2, X2 parts)
symbol range = w1             ; 16 bit word variable for range

main:
    pulsout SIG,2             ; produce 20uS trigger pulse (must be minimum of 10uS)
    pulsin SIG,1,range        ; measures the range in 10uS steps
; now convert range to cm (divide by 5.8) or inches (divide by 14.8)
; as picaxe cannot use 5.8, multiply by 10 then divide by 58 instead
    let range = range * 10 / 58 ; multiply by 10 then divide by 58
    debug range               ; display range via debug command
    pause 50                  ; short delay
    goto main                 ; and around forever

; Note that X2 parts operate at 8MHz instead of 4MHz and so modify the calculation
; let range = range * 5 / 58  ; multiply by (10/2 = 5) then divide by 58
```

Dual Pin Mode - separate trigger / echo microcontroller pins:

The dual pin mode is used for older PICAXE chips such as the 18X or 28X1.
The SRF005 ultrasonic range finder has 5 connections pins. The 3 pin connector is not used in dual pin mode.

Using the 5 way header (note +5V and 0V are marked on the SRF005):

+5V	Connect to 5V
Echo	Connect directly to PICAXE input pin
Trigger	Connect directly to PICAXE output pin
Mode	Do not connect
0V	Connect to 0V

Important - Note that the 'Mode' (pin 4) connection **MUST NOT** be connected for correct operation in this separate trigger/echo mode.

Take care not to overheat, and therefore damage, the solder connection pads whilst making connections.

The SRF005 **Echo Output** is connected to a PICAXE **input** pin.
The SRF005 **Trigger Input** is connected to a PICAXE **output** pin. Note this must be a direct connection to the PICAXE chip leg (do not connect via a darlington driver buffered output on a PICAXE project board).

The following program gives an example of how to use the SRF005 module with a PICAXE microcontroller. Output 3 is used to trigger the SRF005 module via a 'pulsout' command. The SRF005 module then sends out the sonic burst, and sets the Echo Output connection high for the time it takes the sonic burst to be returned. Therefore the PICAXE input (input 6) is used to receive and time this echo pulse via a 'pulsin' command.



The length of the echo pulse is then divided by 5.8 to give a value in cm, and displayed on the computer screen via the 'debug' command. Note that a word variable, w1, is used for the echo timing, as the echo pulse may be a value greater than 255 (maximum value of a byte variable). Word variables are made up of two byte variables and so have a maximum value of 65535 (in this case w1 is made up of b2 and b3, so these two byte variables must not be used anywhere else in the program).

Sample Dual Pin Mode PICAXE Program:

```
symbol trig = 3          ; Define output pin for Trigger pulse (A, M, X, X1 parts)
; symbol trig = b.3      ; Define output pin for Trigger pulse (M2, X2 parts)
symbol echo  = 6         ; Define input pin for Echo pulse (A, M, X, X1 parts)
; symbol echo  = c.6     ; Define input pin for Echo pulse (M2, X2 parts)
symbol range = w1        ; 16 bit word variable for range

main:
   pulsout trig,2        ; produce 20uS trigger pulse (must be minimum of 10uS)
   pulsin echo,1,range   ; measures the range in 10uS steps
   pause 20              ; recharge period after ranging completes
; now convert range to cm (divide by 5.8) or inches (divide by 14.8)
; as picaxe cannot use 5.8, multiply by 10 then divide by 58 instead
   let range = range * 10 / 58       ; multiply by 10 then divide by 58
   debug range                       ; display range via debug command
   goto main                         ; and around forever

; Note that X2 parts operate at 8MHz instead of 4MHz and so modify the calculation
; let range = range * 5 / 58  ; multiply by (10/2 = 5) then divide by 58
```

Example Logicator Flowsheet:

The following flowchart give an example of how to use the SRF005 module with a PICAXE microcontroller in dual pin mode. The special 'ultra' cell is designed for use with the SRF005 and will automatically enable dual pin mode for those PICAXE chips that require it.